(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx opt
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  "NOTE: cannot read -2^62"
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setq minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48)
                              (* 10 (the (integer 0 #.(floor most-positive-fixnum 10))
                                         result))))
              (return (if minus (- result) result))))))))

(in-package :cl-user)

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(declaim (inline calc-prob))
(defun calc-prob (prev-score new-score temp)
  (handler-bind ((floating-point-overflow (lambda (c) #>c (return-from calc-prob 1d0))))
    (exp (/ (- new-score prev-score) temp))))

(defun initialize-scores (d cs ss state)
  (declare (fixnum d)
           ((simple-array uint8 (*)) cs state)
           ((simple-array uint32 (* *)) ss))
  (let* ((prevs (make-array 26 :element-type 'int32 :initial-element -1))
         (res (make-array 26 :element-type 'int32 :initial-element 0)))
    (dotimes (i d)
      (let ((type (aref state i)))
        (incf (aref res type) (aref ss i type))
        (setf (aref prevs type) i)
        (dotimes (j 26)
          (decf (aref res j) (* (aref cs j) (- i (aref prevs j)))))))
    res))

(defun update-scores! (d cs ss state scores old-type new-type)
  (declare #.OPT
           (uint31 d)
           (fixnum old-type new-type)
           ((simple-array uint8 (*)) cs state)
           ((simple-array uint32 (365 26)) ss)
           ((simple-array int32 (*)) scores))
  (setf (aref scores old-type) 0
        (aref scores new-type) 0)
  (let* ((old-prev -1)
         (new-prev -1))
    (declare (int16 old-prev new-prev))
    (dotimes (i d)
      (let ((type (aref state i)))
        (cond ((= type old-type)
               (incf (aref scores type) (aref ss i type))
               (setq old-prev i)
               (decf (aref scores new-type) (* (aref cs new-type) (- i new-prev))))
              ((= type new-type)
               (incf (aref scores type) (aref ss i type))
               (setq new-prev i)
               (decf (aref scores old-type) (* (aref cs old-type) (- i old-prev))))
              (t (decf (aref scores old-type) (* (aref cs old-type) (- i old-prev)))
                 (decf (aref scores new-type) (* (aref cs new-type) (- i new-prev)))))))
    scores))

(defun solve (d cs ss init-state &optional (time 1.93d0))
  (declare #.OPT
           (double-float time)
           (uint32 d)
           ((simple-array uint8 (*)) cs init-state)
           ((simple-array uint32 (* *)) ss))
  (labels ((calc (scores) (loop for x across scores sum x of-type fixnum)))
    (let* ((state (copy-seq init-state))
           (scores (initialize-scores d cs ss state))
           (score (calc scores))
           (opt-scores (copy-seq scores))
           (opt-score (calc opt-scores))
           (opt-state (copy-seq state))
           (start-temp 1000d0)
           (end-temp 1d0)
           (loop-count 0))
      (declare (fixnum loop-count)
               (fixnum opt-score)
               ((simple-array int32 (*)) scores opt-scores)
               (double-float start-temp end-temp))
      (sb-int:with-progressive-timeout (rem-time :seconds time)
        (loop
          (incf loop-count)
          (let ((rem-time (float (rem-time) 1d0)))
            (when (zerop rem-time)
              (return))
            (let* ((day (random d))
                   (old-type (aref state day))
                   (new-type (random 26))
                   (old-score score))
              (unless (= old-type new-type)
                ;; tmp. update
                (setf (aref state day) new-type)
                (update-scores! d cs ss state scores old-type new-type)
                (let* ((new-score (calc scores))
                       (temp (+ end-temp (/ (* rem-time (- start-temp end-temp)) time)))
                       (prob (calc-prob old-score new-score temp)))
                  (declare (double-float temp))
                  ;; rollback
                  (when (> new-score opt-score)
                    (setq opt-scores (copy-seq scores)
                          opt-score new-score
                          opt-state (copy-seq state)))
                  (if (< (random 1.0) prob)
                      (setq score new-score)
                      (progn
                        (setf (aref state day) old-type)
                        (update-scores! d cs ss state scores new-type old-type)))))))))
      #>loop-count
      ;; (assert (= opt-score (calc (initialize-scores d cs ss opt-state))))
      (values opt-state opt-scores))))

(defun print* (state)
  (write-string
   (with-output-to-string (*standard-output* nil :element-type 'base-char)
     (loop for x across state
           do (println (+ x 1))))))

(defun main ()
  (let* ((d (read))
         (cs (make-array 26 :element-type 'uint8 :initial-element 0))
         (ss (make-array (list d 26) :element-type 'uint32 :initial-element 0)))
    (dotimes (i 26)
      (setf (aref cs i) (read-fixnum)))
    (dotimes (i d)
      (dotimes (j 26)
        (setf (aref ss i j) (read-fixnum))))
    ;; #>ss
    (let* ((prevs (make-array 26 :element-type 'int32 :initial-element -1))
           (init-state (make-array d :element-type 'uint8 :initial-element 0)))
      (dotimes (i d)
        (labels ((%eval (type)
                   (let ((value (aref ss i type)))
                     (dotimes (j 26)
                       (unless (= j type)
                         (decf value (* (aref cs j) (- i (aref prevs j))))))
                     value)))
          (let ((max-type 0)
                (max-value most-negative-fixnum))
            (dotimes (type 26)
              (let ((value (%eval type)))
                (when (> value max-value)
                  (setq max-type type
                        max-value value))))
            (setf (aref prevs max-type) i
                  (aref init-state i) max-type))))
      (multiple-value-bind (state scores) (solve d cs ss init-state)
        (print* state)
        (loop for x across scores sum x)))))

#-swank (main)

(defun bench ()
  (let ((*random-state* (seed-random-state 0)))
    (loop repeat 5
          sum #>(run *dat-pathname*))))
;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "365~%")
    (dotimes (i 26)
      (println (random 101) out))
    (dotimes (i 365)
      (dotimes (j 26)
        (println (random 20001) out)))))


Measured memory_bytes: 36888